



VBA voor Doe het Zelfers deel 14

Handleiding van Helpmij.nl

Auteur: leofact

Januari 2015

“ Dé grootste en gratis computerhelpdesk van Nederland ”

Vorige aflevering

In de vorige aflevering werd het maken van een afspraak in de Outlook-agenda met behulp van Excel-VBA besproken. Hiervoor was het nodig dat er vanuit Excel contact werd gelegd met Outlook. In deze aflevering werd ingegaan op de manier waarop vanuit de ene Office-toepassing verbinding kan worden gelegd met een andere Office-toepassing. Dit kun je [hier](#) nalezen.

In deze aflevering


Dit keer wordt er ingegaan op het automatiseren van Word vanuit Excel. Ook hier wordt gebruik gemaakt van Early Binding en Late Binding. Voordat ik daarmee begin wordt er eerst een zijstap gemaakt naar een recente office-update. De bijlage van deze aflevering is [hier](#) te downloaden.

Office-Update


Op 13 december (een zaterdag, dus dat hoeft geen ongeluk te brengen zou je denken) was ik in mijn werkomgeving bezig om een fout te halen uit een werkboek met VBA en ActiveX. Dit wilde ik thuis nog even controleren. Even mailen; openen en ja hoor een foutmelding. Er zou een object missen. Geen van de knoppen, comboboxen etc. werkte. Na vruchteloos speuren naar wat ik nu fout had gedaan ben ik gaan googlen en kwam tot de conclusie dat de fout werd veroorzaakt door een Office-update welke op 9 december door Microsoft is uitgebracht. Deze blijkt een zeer vervelende bijwerking te hebben. Alle ActiveX-elementen worden namelijk in één klap onklaar gemaakt. Bij het openen zal VBA dit melden met de foutmelding "Object is missing". Klikken op knoppen, tekstboxen, selectieboxen en uitklaplijsten geeft geen enkel resultaat. Ook het toevoegen van nieuwe ActiveX-elementen is niet mogelijk; bij een poging hiertoe zal de foutmelding "Cannot insert object" verschijnen. Het blijkt om [deze beveiligingsupdate](#) te gaan.

Het is verbazingwekkend dat deze update door de foutcontrole is gekomen. Er wordt gewerkt aan een oplossing, maar deze is nog niet uitgebracht.

Gelukkig is het probleem zelf wél op te lossen. Hiervoor is het nodig dat je beheer-rechten hebt op het account waarvoor je de aanpassingen wilt gaan doen ([meer informatie](#)). Zorg er vervolgens voor dat verborgen systeembestanden zichtbaar zijn. Dat doe je door in het **Configuratiescherm** bij **Mapopties** > tab **Weergave** > het vinkje weg te halen bij **Beveiligde besturingssysteembestanden verbergen (aanbevolen)**.



Sluit nu eerst alle aanwezige Office-toepassingen. Zoek vervolgens naar alle bestanden met de naam MSForms.exd. In Windows 8.1 doe je dat door de verkenner te openen en dan de Systeem- of Lokale schijf te selecteren (meestal is dat de C-schijf). Vervolgens klik je in het zoekvak rechtsboven. Hierop opent de tab **Hulpprogramma's voor zoeken**. Zorg dat er op **onderliggende mappen** wordt gezocht [1] en dat bij **Geavanceerde opties** de optie **streefbestanden** is geselecteerd [2].



Type hierna MSForms.exd in het zoekvak en verwijder alle resultaten. Zoek in ieder geval naar de volgende bestanden:

C:/Users/Gebruikersnaam/AppData/Local/Temp/Excel8.0/MSforms.exd
C:/Users/Gebruikersnaam/AppData/Local/Temp/VBE/MSforms.exd
C:/Users/Gebruikersnaam/AppData/Local/Temp/Word8.0/MSforms.exd
C:/Users/Gebruikersnaam/AppData/Local/Temp/PPT11.0/MSforms.exd

MSForms.exd is niet altijd op al deze locaties aanwezig.

Aangezien Windows-installaties kunnen verschillen kun je ook op de volgende manier zoeken:

- Druk *Windowstoets* + *R* (voor het uitvoeren vak).
- Type *%temp%* en druk *Enter* of klik **Ok**.

De juiste tijdelijke map wordt nu geopend en hierin kun je naar de genoemde mappen en bestanden zoeken. Het wordt in sommige fora aangeraden om de PC na het verwijderen van MSForms.exd opnieuw op te starten. Bij mij was dat niet nodig. Na het verwijderen van de bestanden werkten alle ActiveX-elementen gelukkig weer zoals ik gewend ben.

Dit blijkt helaas niet altijd zo eenvoudig te gaan. Lukt het niet in één keer dan kun je nog de volgende stappen proberen:

- Open het werkboek met de fout.
- Open de VBA Editor (met *Alt* + *F11*).
- Verander wat in de code. Een commentaarregeltje toevoegen is al voldoende.
- Ga naar menu **Foutopsporing** en klik op **VBAProject compileren**.
- Sla het werkboek op, sluit het af en open het opnieuw.

Met een beetje geluk zou het nu moeten werken. Zo niet, dan moet er mogelijk naar een andere oorzaak worden gezocht. Een overweging kan dan nog zijn om de updates te verwijderen of om via systeemherstel terug te gaan naar een punt, vlak voordat de updates van 9-12 werden geïnstalleerd.

Meer informatie over de beschreven probleemoplossingen is onder andere [in dit Technet blog](#) te vinden. Dit blog zal er ook melding van maken wanneer er een update uit wordt gebracht die bovenstaand probleem oplost. Hopelijk hoeft je daar, na het lezen van dit artikel, niet op te wachten.

Het automatiseren van Word

Nu alles weer werkt (daar gaan we gemakshalve maar even vanuit) gaan we over tot de orde van de dag. Dat is dan een start maken met het automatiseren van Word vanuit Excel.


Afdruk samenvoegen

Wanneer er een afdoende standaard Office-functie bestaat is het aan te bevelen om deze te gebruiken in plaats van zelf aan de gang te gaan met VBA. Om gegevens vanuit Excel in Word te krijgen is dit bijvoorbeeld de functie **Afdruk samenvoegen**. Uitleg van deze functie is onder andere [hier](#) te vinden. Kom je er niet met de ingebouwde functie, dan is het mogelijk om Word te automatiseren vanuit Excel.

Automatiseren

Hiervoor is het nodig dat vanuit Excel een verbinding wordt gelegd met de Word VBA-bibliotheek. We gaan hiervoor eerst een Early Binding leggen zodat we de beschikking hebben over alle programmeerhulpmiddelen waarover de VBA Editor beschikt. Daarna zetten we de Early Binding om in een Late Binding. Hiermee wordt het eventuele verspreiden van het werkboek naar andere computers veel gemakkelijker. Dit proces is in Deel 13 uitgebreid aan de orde geweest voor een verbinding naar Outlook. Voor Word werkt dit in grote lijnen hetzelfde.

De eerste stap is om in de VBA editor de verwijzing te leggen in het **Verwijzingen venster** via menu **Extra > Verwijzingen**. Vink hier de aanwezige Microsoft Word Object Library aan. In dit geval 15.0 voor Word 2013:



De volgende stap is om de verbinding in de procedure te leggen door middel van het declareren van de benodigde objecten. De procedure kan in de basis als volgt worden opgebouwd:

```

Option Explicit

' Procedure : MaakTabel_EarlyBinding1 19-12-2014
' Doel : Bsis voorbeeld Early Binding
'
'
Sub MaakTabel_EarlyBinding1 ()

Dim WordApp As Word.Application
Dim WordDoc As Word.Document
Dim WordTable As Word.Table

'Open de Word-applicatie
Set WordApp = New Word.Application
'Maak een nieuw document
Set WordDoc = WordApp.Documents.Add
'maak het geopende document zichtbaar.
WordApp.Visible = True
End Sub
    
```

Deze procedure maakt een nieuwe Word-instantie aan. Dat wil zeggen dat Word geopend wordt. Dit gebeurt voor een tweede keer als Word al geopend was. Er draaien dan twee "Word-instanties" naast elkaar. Nadat Word is geopend wordt er door de procedure een nieuw document geopend.

Het is eenvoudig om deze procedure om te zetten naar een Late Binding. Net als bij Outlook gebruiken we hier CreateObject voor. De basisprocedure kan er dan als volgt uitzien:

```

' Procedure : MaakTabel_LateBinding1 19-12-2014
' Doel : basis voorbeeld Late Binding
'
'
Sub MaakTabel_LateBinding1 ()

Dim WordApp As Object
Dim WordDoc As Object

'Open de Word-applicatie
Set WordApp = CreateObject("Word.Application")
'Maak een nieuw document
Set WordDoc = WordApp.Documents.Add
'maak het geopende document zichtbaar.
WordApp.Visible = True
End Sub
    
```

Deze procedure werkt nog prima wanneer je de verwijzing weghaalt uit het eerder genoemde venster. Ook wanneer er Word 2010 of eerder geïnstalleerd is op de PC in plaats van Word 2013. De eerste procedure zal in deze gevallen de foutmelding geven dat er een Object gemist wordt.

Tabel

We gaan deze technieken nu inzetten om een procedure te schrijven die een tabel in Word aanmaakt. In deze tabel worden de gegevens van het blad Data opgenomen. De tabel laten we een bepaalde

opmaak geven waarmee nog niet betaalde en te laat betaalde facturen duidelijk herkenbaar worden gemaakt. Bij het programmeren kunnen we IntelliSense, de Object browser, de VBA help en natuurlijk Google (of Bing, DuckDuckGo, of welke zoekmachine je maar wilt) gebruiken. De uitleg van de gebruikte objecten en eigenschappen wordt als commentaar bijgevoegd in de volgende afbeelding van de procedure-code:

Let op! Deze code is in drie delen geknipt omdat de code anders niet op een pagina past.

```
(Algemeen) MaakTabel_EarlyBinding
Option Explicit

' Procedure : MaakTabel_EarlyBinding 19-12-2014
' Doel : Maak een tabel in Word
'
Sub MaakTabel_EarlyBinding()
Dim WordApp As Word.Application
Dim WordDoc As Word.Document
Dim WordTable As Word.Table
Dim WordPar As Paragraph
Dim X As Long
Dim Y As Long
Dim lRij As Long
Dim sTemp As String
Dim WordPar As Paragraph

'open de Word-applicatie
Set WordApp = New Word.Application
'Maak een nieuw document
Set WordDoc = WordApp.Documents.Add
'maak een tabel
With WordDoc
'geef het standaardfont
.Range.Font.Name = "Tahoma"
'geef de inleidende tekst.
.Range(0, 0).Text = "Facturen overzicht van " _
& Strings.Format(Now, "ddd dd - mmmm yyyy", 2) & vbCrLf
'creëer een paragraaf aan om de tabel te positioneren.
Set WordPar = .Paragraphs.Add(.Paragraphs(2).Range)
'maak de eerste rij van de tabel
Set WordTable = .Tables.Add(WordPar.Range, 1, 5)
End With
```

```
'zet de rijhoogte op exact
With WordTable
.Rows.SetHeight 16, wdRowHeightExactly
'vul de eerste vijf cellen
For X = 1 To 5
'vul de eerste vijf cellen één voor één
.Cell(1, X).Range.Text = Cells(15, X + 1)
'geef de celarceping
.Cell(1, X).Shading.BackgroundPatternColor _
= RGB(97, 134, 167)

'geef het font de witte kleur
.Cell(1, X).Range.Font.ColorIndex = wdwhite
'maak de koppen Bold
.Cell(1, X).Range.Font.Bold = True
Next X
'zet de Y variable voor de tabelrij
Y = 2
'loop door de gegevens van het Facturen blad
For lRij = 16 To 40
'voor een nieuwe tabelrij toe
.Rows.Add
'loop langs de 5 cellen in de rij
For X = 1 To 5
With .Cell(Y, X).Range
If lRij Mod 2 = 1 Then
'Arceer de cellen van een oneven rij met de RGB kleur
WordTable.Cell(Y, X).Shading.BackgroundPatternColor _
= RGB(190, 205, 219)

Else
'maak de arceering standaard bij even rijen
WordTable.Cell(Y, X).Shading.BackgroundPatternColorIndex _
= wdAuto

End If
'vul sTemp met de celwaarde uit het Facturen blad
sTemp = Cells(lRij, X + 1)
'zet de tekst standaard op grijs, niet Bold of Italic
.Font.ColorIndex = wdGray50
.Font.Bold = False
.Font.Italic = False
If Cells(lRij, 6) = "te laat" Then
```

```

'maak de tekst rood indien te laat
.Font.ColorIndex = wdRed
ElseIf Cells(lRij, 6) = "nee" Then
'maak de tekst blauw en Italic indien nog niet betaald.
.Font.ColorIndex = wdDarkBlue
.Font.Italic = True
End If
'check of het een datum is en geef juiste format
If IsDate(sTemp) Then sTemp =
Strings.Format(sTemp, "ddd dd - mmm yyyy")
'Check of het een getal is en geef juiste format.
If IsNumeric(sTemp) Then sTemp =
Strings.Format(sTemp, "€ #,00")
'vul de cel met de tekst
.Text = sTemp
End With
'volgende cel
Next X
'volgende rij
Y = Y + 1
Next lRij
'stel de borders in
.Borders.OutsideLineStyle = wdLineStyleSingle
.Borders.OutsideLineWidth = wdLineWidth225pt
.Borders.OutsideColor = RGB(97, 134, 167)
End With
'maak het bestand zichtbaar
WordApp.Visible = True
'reset de Object variabelen
Set WordApp = Nothing
Set WordDoc = Nothing
Set WordTable = Nothing
Set WordPar = Nothing

End Sub

```

De code zelf is in de bijlage terug te vinden. Dat maakt het zelf experimenteren eenvoudiger. De volgende opmaak wordt geprogrammeerd:

- Facturen die betaald zijn worden grijs gemaakt. Hier hoeft namelijk niets mee te gebeuren.
- Facturen die nog betaald moeten worden zijn te herkennen aan de blauwe, gedeeltelijk cursieve tekst.
- De te laat betaalde facturen springen eruit door de rode tekst.

Dat ziet er dan zo uit:

naam	factuur nummer	bedrag	datum	betaald
Pietersen, A.	2014-12-213	€ 750	di 01 - apr 2014	te laat
Kruiningen, W.	2014-12-222	€ 725	za 05 - apr 2014	ja
Jong, D. de	2014-12-231	€ 700	wo 09 - apr 2014	ja
Groen, E.	2014-12-240	€ 675	zo 13 - apr 2014	ja
Boer, R. de	2014-12-249	€ 650	do 17 - apr 2014	te laat
Willemsse, K.	2014-12-258	€ 625	ma 21 - apr 2014	te laat
Geels, O.	2014-12-267	€ 600	vr 25 - apr 2014	te laat
Oort, K. van	2014-12-276	€ 575	di 29 - apr 2014	ja
Wester, F.	2014-12-285	€ 550	za 03 - mei 2014	ja
Wijk, K. van	2014-12-294	€ 525	wo 07 - mei 2014	nee
Droog, K.	2014-12-303	€ 500	zo 11 - mei 2014	ja
Huizing, I.	2014-12-312	€ 475	do 15 - mei 2014	ja
Mulder, S	2014-12-321	€ 450	ma 19 - mei 2014	te laat
Dorsten, I. van	2014-12-330	€ 425	vr 23 - mei 2014	te laat
Blauw, P.	2014-12-339	€ 400	di 27 - mei 2014	te laat
Peters, K.	2014-12-348	€ 375	za 31 - mei 2014	ja
Koopman, L.	2014-12-357	€ 350	wo 04 - jun 2014	nee
Klerk, I.L.	2014-12-366	€ 325	zo 08 - jun 2014	nee
Bos, P.	2014-12-375	€ 300	do 12 - jun 2014	ja
Haanen, L.	2014-12-384	€ 275	ma 16 - jun 2014	nee
Wagenaar -	2014-12-393	€ 250	vr 20 - jun 2014	ja
Smit, F	2014-12-402	€ 225	di 24 - jun 2014	ja
Bouwer, J.	2014-12-411	€ 200	za 28 - jun 2014	nee
Broersen, T	2014-12-420	€ 175	wo 02 - jul 2014	nee
Gerards, K.	2014-12-429	€ 150	zo 06 - jul 2014	te laat

Deze opmaak is natuurlijk maar een voorbeeld. Alles kan naar eigen inzicht worden ingericht.

Late Binding

Mocht het werkboek voor verspreiding bedoeld zijn, dan is het aan te bevelen om de procedure om te zetten naar een Late Binding. Dat gaat net als in het voorbeeld van de basis-procedure. Nu moet er echter wel voor gezorgd worden dat de gebruikte constanten omgezet worden in het getal waar ze voor staan. Doe dit voordat de Early Binding verbroken wordt. Eén van de methodes om de waarden van de constanten te achterhalen is door met de muis boven de constante te zweven (“hoveren”) totdat de waarde zichtbaar wordt.

```
.Rows.SetHeight 16, wdRowHeightExactly
'Vul de eerste vijf cellen met de waarde 2
For X = 1 To 5
```

Omzetten van alle constanten is nu eenvoudig uit te voeren. Het resultaat ziet er als volgt uit:

Let op! Deze code is in drie delen geknipt omdat de code anders niet op een pagina past.

```
(Algemeen) MaakTabel_LateBinding
Option Explicit
' Procedure : MaakTabel_LateBinding 19-12-2014
' Doel : Maak een tabel in Word
'
Sub MaakTabel_LateBinding()
Dim WordApp As Object
Dim WordDoc As Object
Dim WordPar As Object
Dim WordTable As Object
Dim X As Long
Dim Y As Long
Dim lRij As Long
Dim sTemp As String

'Open de Word-applicatie
Set WordApp = CreateObject("Word.Application")
'maak een nieuw document
Set WordDoc = WordApp.Documents.Add
'maak een tabel
With WordDoc
    'Geef het standaardfont:
    .Range.Font.Name = "Tahoma"
    'geef de inleidende tekst.
    .Range(0, 0).Text = "Facturen overzicht van " & _
        & Strings.Format(Now, "ddd dd - mmmm yyyy", 2) & vbCrLf
    'creëer een paragraaf aan om de tabel te positioneren.
    Set WordPar = .Paragraphs.Add(.Paragraphs(2).Range)
    'maak de eerste rij van de tabel
    Set WordTable = .Tables.Add(WordPar.Range, 1, 5)
End With
```



```

'zet de rijkhoogte op exact
With WordTable
.Rows.SetHeight 16, 2
'vul de eerste vijf cellen
For X = 1 To 5
'vul de eerste vijf cellen één voor één.
.Cell(1, X).Range.Text = Cells(15, X + 1)
'geef de celarcering
.Cell(1, X).Shading.BackgroundPatternColor =
    = RGB(97, 134, 167)

'geef het fot de witte kleur
.Cell(1, X).Range.Font.ColorIndex = 8
'maak de koppen Bold
.Cell(1, X).Range.Font.Bold = True
Next X
'zet de Y variable voor de tabelrij
Y = 2
'loop door de gegevens van het Facturen blad
For lRij = 16 To 40
'voor een nieuwe tabelrij toe
.Rows.Add
'loop langs de 5 cellen in de rij
For X = 1 To 5
With .Cell(Y, X).Range

If lRij Mod 2 = 1 Then
'arceer de cellen van een oneven rij met de RGB kleur
WordTable.Cell(Y, X).Shading.BackgroundPatternColor =
    = RGB(190, 205, 219)

Else
'maak de arcering standaard bij even rijen
WordTable.Cell(Y, X).Shading.BackgroundPatternColorIndex =

End If

'vul sTemp met de celwaarde uit het Facturen blad
sTemp = Cells(lRij, X + 1)
'zet de tekst standaard op grijs, niet Bold of Italic.
.Font.ColorIndex = 15
.Font.Bold = False
.Font.Italic = False
If Cells(lRij, 6) = "te laat" Then
'maak de tekst rood indien te laat.
.Font.ColorIndex = 6
ElseIf Cells(lRij, 6) = "nee" Then
'Maak de tekst blauw en Italic indien nog niet betaald.
.Font.ColorIndex = 9
.Font.Italic = True
End If
'check of het een datum is en geef juiste format
If IsDate(sTemp) Then sTemp =
    Strings.Format(sTemp, "ddd dd - mmm yyyy")
'check of het een getal is en geef juiste format.
If IsNumeric(sTemp) Then sTemp =
    Strings.Format(sTemp, "€ #,00")
'vul de cel met de tekst
.Text = sTemp
End With
'Volgende cel
Next X
'Volgende rij
Y = Y + 1
Next lRij
'stel de borders in.
.Borders.OutsideLineStyle = 1
.Borders.OutsideLineWidth = 18
.Borders.OutsideColor = RGB(97, 134, 167)
End With
'maak het bestand zichtbaar
WordApp.Visible = True
'reset de Object variabelen
Set WordApp = Nothing
Set WordDoc = Nothing
Set WordTable = Nothing
Set WordPar = Nothing

End Sub

```

Ook deze code is in de bijlage terug te vinden.

Met deze laatste opmerking komt er een einde aan deze eerste aflevering van 2015.

Samenvatting

Er werd gestart met een waarschuwing voor de Microsoft-Office beveiligingsupdate van 9 december 2014 welke mogelijk alle ActiveX-elementen onbruikbaar maakt. Er werden verschillende methodes beschreven om toch weer met ActiveX te kunnen werken. Daarnaast werd het automatiseren van Word vanuit Excel uitgelegd in Early en Late Binding. Dit aan de hand van een voorbeeld-procedure waarin een tabel met factuur-gegevens werd aangemaakt. De code van deze procedures werd weer in een bijlage bijgeleverd.

Volgende aflevering

Deel 15 zal ingaan op de verschillende mogelijkheden die ons ten dienste staan in VBA om binnen werkboeken naar gegevens te zoeken.

Helpmij.nl